\>

The Differences between Torsion_Helicity and Spin in Electromagnetic Systems.

R.M. Kiehn
Physics Department
University of Houston
August 29,1998
jjkrmk@swbell.net

difsptor.mws

```
> restart:with(plots):with(DEtools):with(linalg):with(difforms):
  with(liesymm):with(plots):setup(x,y,z,t,s,r):defform(x=0,y=0,z=0,r=0,t=0,s=0,a=c
  onst,b=const,c=const,k=const,mu=const,omega=const,m=const):
Warning, new definition for norm
Warning, new definition for trace
Warning, new definition for `&^`
Warning, new definition for close
Warning, new definition for d
Warning, new definition for mixpar
Warning, new definition for translate
Warning, new definition for wdegree
```

The syllabus of classical electromagnetism taught just before and after WWII often contained (and still contains) the dogma that electromagnetic potentials are just a mathematical convenience, and are without physical content. The reasoning is that a relationship between E and B fields and a set of potentials is not unique. Many inequivalent potentials {A, phi} produce the same E and B fields. In fact, there exists a large class of potentials that produce no E and B fields! A particular choice of a set of potentials is said to be a choice of "gauge", and physical theories that are admissable to the syllabus are required to be "gauge invariant", primarily because of the fetish for unique prediction.

In the 50's, the Bohm-Aharanov concept indicated that, at least in "microscopic quantum theory", there could be direct interaction and physical effects due to the potentials -- even in domains where there were no E and B fields. However, even today, there is a resistance to the concept that potentials have physical content in the "macroscopic" domain. Very few physical "theories" contain formulas with explicit utilization of the vector potentials, A. The notable exception is in plasma physics, where attention has focused on the concept defined as helicity density, A dot B. Yet, it is the potentials that contain much of the topological features of the electromagnetic system.

It is now recognized that the electromagnetic field intensities, E and B, are components of a second rank covariant tensor field. In terms of a differential form representation, the 2-form F of field intensities, if closed, is known to have a limit set which replicates the Maxwell-Faraday partial differential equations. However, if the 2-form F is not just closed, but also is exact, such that F-dA=0 globally, then this topological constraint becomes part of the definition of a "classical" electromagnetic system, where the potentials can have physical and topological significance. This topological constraint, F-dA=0 , defined as the Postulate of Potentials, is subsumed herein. Such a topological constraint implies that domains of support for the field intensities, E and B, cannot be compact without boundary, except on the 2D torus or Klein bottle.

Herein, a classical electromagnetic system will be defined by the two topological constraints, F-dA = 0 for the field intensities, E and B, and J-dG=0 for the quantities of excitation, D and H. The field quantities of excitation,D and H, are considered to be distinct from the field intensities, E and B.
Constitutive constraints are considered as additional topological refinements to be added later.

With these assumptions, the classical electromagnetic system admits two topological structures that have explicit dependence upon the potentials:
the Topological Torsion 3-form, A^F,
(whose 4th component is the helicity density)
and the Topological Spin 3-form, A^G.

These concepts of the 3-form of Torsion and the 3-form of Spin are distinct, and have seen almost no utilization in plasma physics, even though the 4th component of A^F is the ubiquitous Helicity density that appears in many studies of MHD.

For details, see
http://www.uh.edu/~rkiehn/pdf/classice.pdf

A few simple examples are presented below to show that, even in stationary domains,
the 3-forms of Torsion and Spin are distinct in most classical electromagnetic systems.
The examples presume certain forms for the vector and scalar potentials, and then
generate the E,B, D,H,J,rho fields according to Maxwell's equations. What physical systems
they represent is not of issue. The examples are to demonstrate solutions to Maxwell's equations for
which the concepts of topological torsion and spin exist and are distinct.

There exist examples where the Spin 3-form is not zero and the Torsion 3-form is zero.
and other examples where the Spin 3-form is zero, but the Torsion 3-form is not zero.

For an interesting model of a rotating time dependent plasma system that yields an
accretion disc see
http://www.uh.edu/~rkiehn/pdf/diracch.pdf

****
When either 3-form is non-zero, it is common to find helical distributions of certain vector lines.

PRELIMINARY

Consider Vector potentials for an electromagnetic field which consist of a factor f{r} times a closed, but not exact, 1-form (the Kelvin 1-form) which has finite circulation, but zero curl almost everywhere. Add to this 1-form, a component in the direction of the rotation axis, times a different functional factor, g(r,z).
The electromagnetic 1-form of Action (time independent, and without a coulomb term) becomes:

```
> Kelvin:=(-y*d(x)+x*d(y))/(y^2+x^2);Action:=f(r,z)*Kelvin -g(r,z)/a*d(z);
```

$$Kelvin := \frac{-y\,\mathrm{d}(x) + x\,\mathrm{d}(y)}{y^2 + x^2}$$

$$Action := \frac{\mathrm{f}(r, z)\,(-y\,\mathrm{d}(x) + x\,\mathrm{d}(y))}{y^2 + x^2} - \frac{\mathrm{g}(r, z)\,\mathrm{d}(z)}{a}$$

A remarkable feature of such an Action 1-form is that it is of Pfaff dimension 2. It does not exhibit helicity. Yet for certain functional forms of f(r,z) the magnetic field is trkalian and generates B field lines that are helical in shape. The B field is such that the induced current is proportional to B such that the Lorentz force term J x B vanishes. Such a field describes what is called the "force-free" plasma. Yet the Helicity density is ZERO.

EXAMPLE 1

Torsion 3-form = Zero, Spin 3-form Not Zero.

Consider the function (from the book by G. Marsh)

```
> f(r):=ln((1)^2+(x/a)^2+(y/a)^2);
```

$$\mathrm{f}(r) := \ln\left(1 + \frac{x^2}{a^2} + \frac{y^2}{a^2}\right)$$

Compute the explicit form of the Action and vector potential. Set g=f.

```
> ATYPE:=alpha*f(r)*Kelvin -beta*f(r)/a*d(z);
  A1:=-alpha*y*f(r)/(x^2+y^2);A2:=alpha*x*f(r)/(x^2+y^2);A3:=-beta*f(r)/a;A4:=0;
```

$$ATYPE := \frac{\alpha \ln\left(1 + \frac{x^2}{a^2} + \frac{y^2}{a^2}\right)(-y\,\mathrm{d}(x) + x\,\mathrm{d}(y))}{y^2 + x^2} - \frac{\beta \ln\left(1 + \frac{x^2}{a^2} + \frac{y^2}{a^2}\right)\mathrm{d}(z)}{a}$$

$$A1 := -\frac{\alpha\, y \ln\left(1 + \frac{x^2}{a^2} + \frac{y^2}{a^2}\right)}{y^2 + x^2}$$

$$A2 := \frac{\alpha\, x \ln\left(1 + \frac{x^2}{a^2} + \frac{y^2}{a^2}\right)}{y^2 + x^2}$$

$$A3 := -\frac{\beta \ln\left(1 + \frac{x^2}{a^2} + \frac{y^2}{a^2}\right)}{a}$$

$$A4 := 0$$

Construct the 2-form of field Intentities:

```
> F:=wcollect(d(ATYPE -A4*d(t)));
```

$$F := \left(-\frac{\alpha\,(\%1\, y^2 a^2 + \%1\, y^4 - \%1\, x^2 a^2 - \%1\, x^4 - 2\, y^4 - 2\, y^2 x^2)}{(y^2 + x^2)^2\,(a^2 + x^2 + y^2)} + \frac{\alpha\,(\%1\, y^2 a^2 + \%1\, y^4 - \%1\, x^2 a^2 - \%1\, x^4 + 2\, y^2 x^2 + 2\, x^4)}{(y^2 + x^2)^2\,(a^2 + x^2 + y^2)}\right)((\mathrm{d}(x))\ \&\wedge\ (\mathrm{d}(y))) - 2\,\frac{\beta\, x\,((\mathrm{d}(x))\ \&\wedge\ (\mathrm{d}(z)))}{a\,(a^2 + x^2 + y^2)}$$

$$-2\,\frac{\beta\,y\,((\mathrm{d}(y))\ \&\wedge\ (\mathrm{d}(z)))}{a\,(a^2+x^2+y^2)}$$

$$\%1 := \ln\left(\frac{a^2+x^2+y^2}{a^2}\right)$$

```
> B1:=getcoeff(wcollect(F&^d(x)&^d(t)));B2:=getcoeff(wcollect(F&^d(y)&^d(t)));B3:=
  factor(getcoeff(wcollect(F&^d(z))));
```

The components of the B field generate a set of field lines that swirl in a helical fashion about the z axis. (The components of the vector potential, A, also form a similar set of helicies)

$$B1 := -2\,\frac{\beta\,y}{a\,(a^2+x^2+y^2)}$$

$$B2 := 2\,\frac{\beta\,x}{a\,(a^2+x^2+y^2)}$$

$$B3 := 2\,\frac{\alpha}{a^2+x^2+y^2}$$

The computations can be done in engineering format to yield the same answers:

```
> APOT1:=([A1,A2,A3]);
```

$$APOT1 := \left[-\frac{\alpha\,y\,\ln\left(1+\frac{x^2}{a^2}+\frac{y^2}{a^2}\right)}{y^2+x^2},\ \frac{\alpha\,x\,\ln\left(1+\frac{x^2}{a^2}+\frac{y^2}{a^2}\right)}{y^2+x^2},\ -\frac{\beta\,\ln\left(1+\frac{x^2}{a^2}+\frac{y^2}{a^2}\right)}{a}\right]$$

```
> Bf1:=curl(APOT1,[x,y,z]):B1:=factor(Bf1[1]);B2:=factor(Bf1[2]);B3:=factor(Bf1[3]
  );
```

$$B1 := -2\,\frac{\beta\,y}{a\,(a^2+x^2+y^2)}$$

$$B2 := 2\,\frac{\beta\,x}{a\,(a^2+x^2+y^2)}$$

$$B3 := 2\,\frac{\alpha}{a^2+x^2+y^2}$$

As the vector potential is presumed to be time independent, and there is no scalar potential, then there is no E field when A4= 0.

```
> E:=[0,0,-diff(A4,z)];
```

$$E := [0, 0, 0]$$

Even though both the lines of B and the lines of A are helical, the Helicity density vanishes

```
> Helicity:=innerprod(APOT1,Bf1);
```

$$Helicity := 0$$

Use the Vacuum consitutive equations B = mu H to construct the Amperian currents

```
> Jf1:=curl(Bf1/mu,[x,y,z]):J1:=factor(Jf1[1]);J2:=factor(Jf1[2]);J3:=factor(Jf1[3
  ]);
```

$$J1 := -4\,\frac{\alpha\,y}{(a^2+x^2+y^2)^2\,\mu}$$

$$J2 := 4\,\frac{\alpha\,x}{(a^2+x^2+y^2)^2\,\mu}$$

$$J3 := 4\,\frac{\beta\,a}{\mu\,(a^2+x^2+y^2)^2}$$

Note that when alpha=plus or minus beta, the Amperian current is proportional to the B field, and the plasma is "FORCE FREE". Fluctuations in alpha and beta would eliminate the "force free" property. Also note that J is proportional to A or B under similar alpha,beta constraints.

```
> B1/J1;B2/J2;B3/J3;JxB:=crossprod(Jf1,Bf1):JxBX:=factor(JxB[1]);JxBY:=factor(JxB[2]);JxBZ:=factor(JxB[3]);
```

$$\frac{1}{2}\frac{\beta\,(a^2+x^2+y^2)\,\mu}{a\,\alpha}$$

$$\frac{1}{2}\frac{\beta\,(a^2+x^2+y^2)\,\mu}{a\,\alpha}$$

$$\frac{1}{2}\frac{\alpha\,(a^2+x^2+y^2)\,\mu}{\beta\,a}$$

$$JxBX := -8\,\frac{(\beta-\alpha)\,(\beta+\alpha)\,x}{\mu\,(a^2+x^2+y^2)^3}$$

$$JxBY := -8\,\frac{(\beta-\alpha)\,(\beta+\alpha)\,y}{\mu\,(a^2+x^2+y^2)^3}$$

$$JxBZ := 0$$

When alpha/beta= plus or minus 1 (left handed or right handed), the B Field is trkalian, J x B = 0, and the Helicity = zero! Yet all Field lines look like helicies !!
As there is no scalar potential, and no E field, The torsion vector is identically zero.

The Magnetic Helicity = J dot B is NOT zero

```
> MagHelicity1:=innerprod(Bf1,Jf1);
```

$$MagHelicity1 := 8\,\frac{\beta\,\alpha}{a\,(a^2+x^2+y^2)^2\,\mu}$$

Magnetic Energy B dot H compared to Interaction energy, A dot J:

```
> MagE:=innerprod(Bf1,Bf1)/mu;
```

$$MagE := 4\,\frac{\beta^2 y^2+\beta^2 x^2+\alpha^2 a^2}{a^2\,(a^2+x^2+y^2)^2\,\mu}$$

```
> IntE:=innerprod(APOT1,Jf1);
```

$$IntE := -4\,\frac{(-\alpha^2+\beta^2)\,\ln\left(\frac{a^2+x^2+y^2}{a^2}\right)}{\mu\,(a^2+x^2+y^2)^2}$$

The Interaction energy vanishes when the rotation is isotropic.
Although the 4 components of the Torsion vector are Zero, the Spin vector is not ZERO !

```
> Spin:=evalm(crossprod(APOT1,Bf1/mu)+A4*epsilon*E):sp1:=factor(Spin[1]);sp2:=factor(Spin[2]);sp3:=factor(Spin[3]);P1:=factor(diverge(Spin,[x,y,z]));
```

The Spin vector is hedgehog in 2D, and is NOT zero.

$$sp1 := 2\,\frac{(\beta^2 y^2+\beta^2 x^2+\alpha^2 a^2)\,x\,\ln\left(\frac{a^2+x^2+y^2}{a^2}\right)}{a^2\,(a^2+x^2+y^2)\,\mu\,(y^2+x^2)}$$

$$sp2 := 2\,\frac{(\beta^2 y^2+\beta^2 x^2+\alpha^2 a^2)\,\ln\left(\frac{a^2+x^2+y^2}{a^2}\right)y}{a^2\,(a^2+x^2+y^2)\,\mu\,(y^2+x^2)}$$

$$sp3 := 0$$

$$P1 := 4\,\frac{\beta^2 y^2 + \alpha^2 a^2 + \beta^2 \ln\left(\frac{a^2+x^2+y^2}{a^2}\right) a^2 + \beta^2 x^2 - \alpha^2 \ln\left(\frac{a^2+x^2+y^2}{a^2}\right) a^2}{a^2 \mu \left(a^2+x^2+y^2\right)^2}$$

The First Poincare invariant, P1, is not zero!.

```
> factor(subs(alpha=beta,P1));
```

$$4\,\frac{\beta^2}{\mu\left(a^2+x^2+y^2\right) a^2}$$

EXAMPLE 2

Torsion 3-form not zero, Spin 3-form = 0.

As a next example consider the function f(r,z) to be specified as:

```
> g2(r):=1/(1+(x/a)^2+(y/a)^2);f2(r):=(x^2+y^2)/(1+(x/a)^2+(y/a)^2);
```

$$\mathrm{g2}(r) := \frac{1}{1+\frac{x^2}{a^2}+\frac{y^2}{a^2}}$$

$$\mathrm{f2}(r) := \frac{y^2+x^2}{1+\frac{x^2}{a^2}+\frac{y^2}{a^2}}$$

The choice for f2 makes the Pfaff dimension of the Action 1-form, AT2, greater than 2.

```
> AT2:=f2(r)*Kelvin -g2(r)*d(z);
  A1:=-y*f2(r)/(x^2+y^2);A2:=x*f2(r)/(x^2+y^2);A3:=-a*g2(r);
>
```

$$AT2 := \frac{-y\,\mathrm{d}(x) + x\,\mathrm{d}(y)}{1+\frac{x^2}{a^2}+\frac{y^2}{a^2}} - \frac{\mathrm{d}(z)}{1+\frac{x^2}{a^2}+\frac{y^2}{a^2}}$$

$$A1 := -\frac{y}{1+\frac{x^2}{a^2}+\frac{y^2}{a^2}}$$

$$A2 := \frac{x}{1+\frac{x^2}{a^2}+\frac{y^2}{a^2}}$$

$$A3 := -\frac{a}{1+\frac{x^2}{a^2}+\frac{y^2}{a^2}}$$

```
> APOT2:=evalm([A1,A2,A3]);
> Bf2:=evalm(curl(APOT2,[x,y,z])):B1:=factor(Bf2[1]);B2:=factor(Bf2[2]);B3:=factor
  (Bf2[3]);Helicity2:=innerprod(APOT2,Bf2);
```

$$APOT2 := \left[-\frac{y}{1+\frac{x^2}{a^2}+\frac{y^2}{a^2}},\ \frac{x}{1+\frac{x^2}{a^2}+\frac{y^2}{a^2}},\ -\frac{a}{1+\frac{x^2}{a^2}+\frac{y^2}{a^2}}\right]$$

$$B1 := 2 \frac{a^3 y}{(a^2 + x^2 + y^2)^2}$$

$$B2 := -2 \frac{a^3 x}{(a^2 + x^2 + y^2)^2}$$

$$B3 := 2 \frac{a^4}{(a^2 + x^2 + y^2)^2}$$

$$Helicity2 := -2 \frac{a^5}{(a^2 + x^2 + y^2)^2}$$

Again BOTH the field lines of B and the field lines of A are helical in a 3D sense. But in this case the Helicity density is NOT zero. In this case, the factor r^2 appears in the denominator of f2, such that the resulting form is not the product of a scalar function and a closed 1-form. Hence the Pfaff dimension is not 2. Helicity only appears when the Pfaff dimension is greater than 2.

Evaluate the cross product of A and B to show that the A field is Trkalian, but the B field is not.

```
> AXB2:=crossprod(APOT2,Bf2):factor(AXB2[1]);factor(AXB2[2]);factor(AXB2[3]);facto
  r(A1/B1);factor(A2/B2);factor(A3/B3);
```

$$0$$

$$0$$

$$0$$

$$-\frac{1}{2} \frac{a^2 + x^2 + y^2}{a}$$

$$-\frac{1}{2} \frac{a^2 + x^2 + y^2}{a}$$

$$-\frac{1}{2} \frac{a^2 + x^2 + y^2}{a}$$

Use the Vacuum consitutive equations B = mu H to construct the Amperian currents

```
> Jf2:=curl([B1,B2,B3]/mu,[x,y,z]):J1:=factor(Jf2[1]);J2:=factor(Jf2[2]);J3:=facto
  r(Jf2[3]);
```

$$J1 := -8 \frac{a^4 y}{\mu (a^2 + x^2 + y^2)^3}$$

$$J2 := 8 \frac{a^4 x}{\mu (a^2 + x^2 + y^2)^3}$$

$$J3 := -4 \frac{a^3 (a^2 - x^2 - y^2)}{\mu (a^2 + x^2 + y^2)^3}$$

Note that the Amperian current is NOT proportional to the B field as it was in the FIRST example
Hence the field is not force free in the sense that J x B is not zero.
The direction field of the currents are helicies and current flows in the positive z direction outside the cylindrical radius, a, and in the negative z direction inside the cylinder of diameter a.

```
> B1/J1;B2/J2;B3/J3;
```

$$-\frac{1}{4} \frac{(a^2 + x^2 + y^2) \mu}{a}$$

$$-\frac{1}{4}\frac{(a^2+x^2+y^2)\,\mu}{a}$$

$$-\frac{1}{2}\frac{a\,(a^2+x^2+y^2)\,\mu}{a^2-x^2-y^2}$$

```
> JxB:=crossprod(Jf2,Bf2):JxBX:=factor(JxB[1]);JxBY:=factor(JxB[2]);JxBZ:=factor(JxB[3]);
```

$$JxBX := 8\frac{x\,a^6}{(a^2+x^2+y^2)^4\,\mu}$$

$$JxBY := 8\frac{y\,a^6}{(a^2+x^2+y^2)^4\,\mu}$$

$$JxBZ := 0$$

The A Field is trkalian, but J x B is not zero, and the Helicity is not zero! The Lorentz force is radially symmetric, and the plasma is NOT force free.

```
> MagHelicity2:=innerprod(Bf2,Jf2);
```

$$MagHelicity2 := -8\frac{a^7}{\mu\,(a^2+x^2+y^2)^4}$$

Magnetic Energy compared to interaction energy.

```
> MagE2:=innerprod(Bf2,Bf2)/mu;
> IntE2:=innerprod(APOT2,Jf2);
```

$$MagE2 := 4\frac{a^6}{(a^2+x^2+y^2)^3\,\mu}$$

$$IntE2 := 4\frac{a^6}{(a^2+x^2+y^2)^3\,\mu}$$

Although the spatial part of the Torsion vector is Zero, the Helicity density (4th component of the Torsion 3-form) is not zero.

However in this example, counter to the first example, the Spin vector is exactly ZERO in all 4 coponents ! The Spin 3-form is indentically zero.

```
> Spin:=crossprod(APOT2,Bf2):sp1:=factor(Spin[1]);sp2:=factor(Spin[2]);sp3:=factor(Spin[3]);
```

$$sp1 := 0$$

$$sp2 := 0$$

$$sp3 := 0$$

```
>
>
```

EXAMPLE 3

As a third example consider the B field HedgeHog solution

(Set beta = delta for a finite scalar potential example)

```
> beta:=0;
>
```

$$\beta := 0$$

```
> f3:=z*(alpha*(cos(omega*t)))/((z)^2+(x)^2+(y)^2)^(1/2);
```

$$f3 := \frac{z\,\alpha\cos(\omega\,t)}{\sqrt{z^2+x^2+y^2}}$$

```
> A1:=-y*f3/(x^2+y^2);A2:=x*f3/(x^2+y^2);A3:=-0;A4:=beta*z*sin(omega*t);
```

$$A1 := -\frac{y\,z\,\alpha\cos(\omega\,t)}{\sqrt{z^2+x^2+y^2}\,(y^2+x^2)}$$

$$A2 := \frac{x\,z\,\alpha\cos(\omega\,t)}{\sqrt{z^2+x^2+y^2}\,(y^2+x^2)}$$

$$A3 := 0$$

$$A4 := 0$$

```
> APOT3:=evalm([A1,A2,A3]);
> Bf3:=evalm(curl(APOT3,[x,y,z])):B1:=factor(Bf3[1]);B2:=factor(Bf3[2]);B3:=factor
  (Bf3[3]);Helicity3:=innerprod(APOT3,Bf3);
```

$$APOT3 := \left[-\frac{y\,z\,\alpha\cos(\omega\,t)}{\sqrt{z^2+x^2+y^2}\,(y^2+x^2)},\ \frac{x\,z\,\alpha\cos(\omega\,t)}{\sqrt{z^2+x^2+y^2}\,(y^2+x^2)},\ 0\right]$$

$$B1 := -\frac{x\,\alpha\cos(\omega\,t)}{(z^2+x^2+y^2)^{3/2}}$$

$$B2 := -\frac{y\,\alpha\cos(\omega\,t)}{(z^2+x^2+y^2)^{3/2}}$$

$$B3 := -\frac{z\,\alpha\cos(\omega\,t)}{(z^2+x^2+y^2)^{3/2}}$$

$$Helicity3 := 0$$

The Helicity density is zero. NOTE THAT THE B FIELD IS HEDGEHOG and is independent from beta.

Evaluate the cross product of A and B to show that the A field is NOT Trkalian

```
> AXB3:=crossprod(APOT3,Bf3):factor(AXB3[1]);factor(AXB3[2]);factor(AXB3[3]);facto
  r(A1/B1);factor(A2/B2);factor(A3/B3);
```

$$-\frac{\cos(\omega\,t)^2\,\alpha^2\,z^2\,x}{(y^2+x^2)\,(z^2+x^2+y^2)^2}$$

$$-\frac{\cos(\omega\,t)^2\,\alpha^2\,z^2\,y}{(y^2+x^2)\,(z^2+x^2+y^2)^2}$$

$$\frac{z\,\alpha^2\cos(\omega\,t)^2}{(z^2+x^2+y^2)^2}$$

$$\frac{y\,z\,(z^2+x^2+y^2)}{(y^2+x^2)\,x}$$

$$-\frac{x\,z\,(z^2+x^2+y^2)}{(y^2+x^2)\,y}$$

$$0$$

Use the Vacuum consitutive equations B = mu H to construct the

Amperian currents

```
> Jf3:=curl([B1,B2,B3]/mu,[x,y,z]):J1:=factor(Jf3[1]);J2:=factor(Jf3[2]);J3:=facto
  r(Jf3[3]);
```

$$J1 := 0$$
$$J2 := 0$$
$$J3 := 0$$

The Amperian currents vanish unless the vector potentials have some time dependence; alpha is not zero. (fluctuations?)

DISPLACEMENT CURRENTS

```
> Ef3:=evalm([diff(-A1,t)-diff(A4,x),diff(-A2,t)-diff(A4,y),diff(-A3,t)-diff(A4,z)
  ]);rho:=diverge(Ef3,[x,y,z]);
```

$$Ef3 := \left[-\frac{y\,z\,\alpha\,\sin(\omega\,t)\,\omega}{\sqrt{z^2+x^2+y^2}\,(y^2+x^2)}, \frac{x\,z\,\alpha\,\sin(\omega\,t)\,\omega}{\sqrt{z^2+x^2+y^2}\,(y^2+x^2)}, 0 \right]$$
$$\rho := 0$$

```
> JDf3:=evalm([diff(Ef3[1],t),diff(Ef3[2],t),diff(Ef3[3],t)]);
```

$$JDf3 := \left[-\frac{y\,z\,\alpha\,\cos(\omega\,t)\,\omega^2}{\sqrt{z^2+x^2+y^2}\,(y^2+x^2)}, \frac{x\,z\,\alpha\,\cos(\omega\,t)\,\omega^2}{\sqrt{z^2+x^2+y^2}\,(y^2+x^2)}, 0 \right]$$

```
> Jtot:=evalm(Jf3-epsilon*JDf3);
```

$$Jtot := \left[\frac{\varepsilon\,y\,z\,\alpha\,\cos(\omega\,t)\,\omega^2}{\sqrt{z^2+x^2+y^2}\,(y^2+x^2)}, -\frac{\varepsilon\,x\,z\,\alpha\,\cos(\omega\,t)\,\omega^2}{\sqrt{z^2+x^2+y^2}\,(y^2+x^2)}, 0 \right]$$

```
> JxB:=crossprod(Jtot,Bf3):XcompLF:=factor(JxB[1]);YcompLF:=factor(JxB[2]);ZcompLF
  :=factor(JxB[3]);
```

$$XcompLF := \frac{\omega^2\,\cos(\omega\,t)^2\,\alpha^2\,z^2\,x\,\varepsilon}{(y^2+x^2)\,(z^2+x^2+y^2)^2}$$
$$YcompLF := \frac{\omega^2\,\cos(\omega\,t)^2\,\alpha^2\,z^2\,y\,\varepsilon}{(y^2+x^2)\,(z^2+x^2+y^2)^2}$$
$$ZcompLF := -\frac{\varepsilon\,z\,\alpha^2\,\cos(\omega\,t)^2\,\omega^2}{(z^2+x^2+y^2)^2}$$

The system is NOT force free for finite alpha, but there are no finite charge densities, rho
The force is radial outbound .
The z component of the force is always directed to the xy plane z = 0 plane.

Such is the form of a force to create an accretion disk in the presence of an attractive central field.

The solution is not a vacuum solution for there are finite current densities, J. But the Helicity density of the Vacuum is ZERO.!

```
> Helicity3:=innerprod(APOT3,Bf3);
```

$$Helicity3 := 0$$

```
>
```

The Magnetic Helicity is zero

```
> MagHelicity3:=innerprod(Bf3,Jf3);
```

$$MagHelicity3 := 0$$

Magnetic Energy B.H compared to interaction energy A.J.

```
> MagE3:=innerprod(Bf3,Bf3)/mu;ElecE3:=epsilon*innerprod(Ef3,Ef3);
```

$$MagE3 := \frac{\alpha^2 \cos(\omega t)^2}{(z^2+x^2+y^2)^2 \mu}$$

$$ElecE3 := \frac{\varepsilon z^2 \alpha^2 \sin(\omega t)^2 \omega^2}{(y^2+x^2)(z^2+x^2+y^2)}$$

```
> IntE3:=innerprod(APOT3,Jtot);
```

$$IntE3 := -\frac{z^2 \alpha^2 \cos(\omega t)^2 \varepsilon \omega^2}{(y^2+x^2)(z^2+x^2+y^2)}$$

```
>
> Tors:=evalm(crossprod(Ef3,APOT3)+A4*Bf3):TORSION:=([factor(Tors[1]),factor(Tors[
  2]),factor(Tors[3]),Helicity3]);EdotB:=factor(innerprod(Ef3,Bf3));P2:=-2*EdotB;
```

$$TORSION := [0, 0, 0, 0]$$

$$EdotB := 0$$

$$P2 := 0$$

The Torsion 3-form is identically zero in the case beta=0, but the Spin 3-form is NOT zero.
The divergence of the Torsion 3-form is zero if beta = 0. But if beta is not zero, then the Torsion 3-form is not zero and the integral of the 3-form of torsion is not conserved, as P2 is not zero.

The Spin 3-form is not zero.

```
> Spin:=evalm(crossprod(APOT3,Bf3/mu)+A4*epsilon*Ef3):sp1:=factor(Spin[1]);sp2:=fa
  ctor(Spin[2]);sp3:=factor(Spin[3]);sp4:=factor(innerprod(APOT3,epsilon*Ef3));SPI
  N:=evalm([sp1,sp2,sp3,sp4]);
```

$$sp1 := -\frac{\cos(\omega t)^2 \alpha^2 z^2 x}{\mu (y^2+x^2)(z^2+x^2+y^2)^2}$$

$$sp2 := -\frac{\cos(\omega t)^2 \alpha^2 z^2 y}{\mu (y^2+x^2)(z^2+x^2+y^2)^2}$$

$$sp3 := \frac{z \alpha^2 \cos(\omega t)^2}{\mu (z^2+x^2+y^2)^2}$$

$$sp4 := \frac{z^2 \alpha^2 \cos(\omega t) \varepsilon \sin(\omega t) \omega}{(y^2+x^2)(z^2+x^2+y^2)}$$

$$SPIN := \left[-\frac{\cos(\omega t)^2 \alpha^2 z^2 x}{\mu (y^2+x^2)(z^2+x^2+y^2)^2}, -\frac{\cos(\omega t)^2 \alpha^2 z^2 y}{\mu (y^2+x^2)(z^2+x^2+y^2)^2}, \frac{z \alpha^2 \cos(\omega t)^2}{\mu (z^2+x^2+y^2)^2}, \frac{z^2 \alpha^2 \cos(\omega t) \varepsilon \sin(\omega t) \omega}{(y^2+x^2)(z^2+x^2+y^2)}\right]$$

Note that the z component of the Spin vector changes sign with z.
The radial components of the Spin are attracted to the origin,
the z components are repelled by the origin.
The Spin density sp4 has twice the oscillation frequency

```
> P1:=factor(simplify(factor(subs(diverge(SPIN,[x,y,z,t]))),trig));
```

$$P1 := \alpha^2 (-\varepsilon z^4 \omega^2 \mu + 2 z^4 \cos(\omega t)^2 \varepsilon \omega^2 \mu - y^2 \varepsilon z^2 \omega^2 \mu + 2 y^2 z^2 \cos(\omega t)^2 \varepsilon \omega^2 \mu + 2 z^2 \cos(\omega t)^2 \varepsilon \omega^2 \mu x^2 - \varepsilon z^2 \omega^2 \mu x^2 + y^2 \cos(\omega t)^2 + \cos(\omega t)^2 x^2) \Big/ (\mu (y^2 + x^2) (z^2 + x^2 + y^2)^2)$$

```
>
> CHEK:=factor(P1-MagE3+ElecE3+IntE3-rho*A4);
```

$$CHEK := \frac{(-1 + \cos(\omega t)^2 + \sin(\omega t)^2)\, \omega^2 z^2 \varepsilon \alpha^2}{(z^2 + x^2 + y^2)(y^2 + x^2)}$$

```
>
```

(CHEK = 0 for computational accuracy.)

The First Poincare invariant is not zero which implies that the integral of the 3-form of SPIN is not conserved.

For more details see http://www.uh.edu/~rkiehn/pdf/diracch.pdf

```
>
```